

Modélisation des isothermes d'Andrews du dioxyde de carbone.

Première partie : modèle de Van der Waals paracritique

Le dioxyde de carbone CO_2 a une température critique $t_c = 31,0^\circ\text{C}$, une pression critique $p_c = 73,9$ bar, le volume molaire critique étant $V_{mc} = 94,0 \text{ cm}^3 \cdot \text{mol}^{-1}$. Nous cherchons à réaliser un modèle rendant compte de certaines propriétés de ce fluide dans des conditions peu éloignées de l'état critique (c'est ce que signifie *paracritique*).

1- Considérons le modèle de Van der Waals à trois paramètres dans lequel la pression p , le volume molaire V_m et la température thermodynamique T sont reliés par l'équation d'état suivante :

$$\left(p + \frac{a}{V_m^2} \right) (V_m - b) = rT$$

À l'aide du logiciel de calcul formel *Maple*, déterminer la valeur des paramètres a , b et r du fluide de Van der Waals ayant le même point critique que CO_2 .

Il suffit d'écrire que sur l'isotherme critique, le point critique est un point d'inflexion de pente nulle,

$$\text{soit pour } T = T_c \text{ et } V_m = V_{mc} : \frac{\partial p(T, V_m)}{\partial V_m}(T_c, V_{mc}) = 0 \quad \text{et} \quad \frac{\partial^2 p(T, V_m)}{\partial V_m^2}(T_c, V_{mc}) = 0$$

```
> p:=r*T/(v-b)-a/v**2;
pc:=73.9e5: Tc:=31+273.15: vc:=94e-6:
T:=Tc:
abr:=solve({subs(v=vc,p)=pc,
             subs(v=vc,diff(p,v))=0,
             subs(v=vc,diff(p,v,v))=0},{a,b,r});
assign(abr);
```

$$p := \frac{rT}{v-b} - \frac{a}{v^2}$$

$$abr := \{ a = 0.1958941199, b = 0.00003133333334, r = 6.090503586 \}$$

La réponse de *Maple* se présente sous la forme d'un ensemble d'équations (type « set ») et l'ordre des variables est tout à fait imprévisible. Cet ensemble d'équations est transformé en ensemble d'assignations de valeurs aux variables a , b et r par l'instruction **assign**.

Remarque : Le physicien lira ces résultats de la façon suivante :

$$a = 0,196 \text{ Pa} \cdot \text{m}^6 \cdot \text{mol}^{-2}, \quad b = 31,3 \times 10^{-6} \text{ m}^3 \cdot \text{mol}^{-1} = 31,3 \text{ mL} \cdot \text{mol}^{-1}, \quad r = 6,09 \text{ J} \cdot \text{K}^{-1} \cdot \text{mol}^{-1}$$


Note : la valeur de r est notablement différente de la valeur de la constante universelle des gaz parfaits $R = 8,31 \text{ J} \cdot \text{K}^{-1} \cdot \text{mol}^{-1}$. Cela n'est pas du tout étonnant : nous décrivons un fluide au voisinage de son état critique, lequel correspond à une forte densité : nous sommes très loin du modèle de gaz parfait.

2- Dans ce modèle de Van der Waals, représenter le réseau d'isothermes d'Andrews du dioxyde de carbone pour une gamme de températures de 25°C à 35°C , de degré en degré. L'isotherme critique sera représentée par un trait plus gras.

L'opérateur **\$** permet, dans le langage *Maple* de construire automatiquement une séquence par incrémentation d'un entier (ici **t**) dont on donne l'amplitude de variation par un intervalle (ici **\$t=25..35**). L'instruction **plot** de *Maple* peut s'appliquer à une liste d'objets, les options pouvant

dans ce cas être définies sous forme d'une liste d'options prenant des valeurs différentes pour chaque objet. C'est ce que réalisent les deux lignes de programme suivantes :


```
> T:=273.15+t;
plot([p $t=25..35],v=0..2*vc,y=0.9*pc..1.1*pc,
color=red,numpoints=200,
thickness=[1$6,3,1$4]);
```



Ce modèle est-il réaliste ?

Le modèle de Van der Waals introduit, pour les températures subcritiques, des états de compressibilité négative que l'on n'observe jamais. Il s'agirait de fluides dont le volume augmente lorsqu'on les soumet à des pressions croissantes : cela n'existe pas dans le monde réel. Le modèle de Van der Waals n'est donc pas réaliste, il devra être modifié pour tenir compte des phénomènes de changement d'état.

- 3- Pour rendre compte des phénomènes de changement d'état, les isothermes subcritiques ($T < T_c$) sont modifiées par l'introduction d'un palier de saturation à la pression de saturation p_s . Les points L_s et V_s représentent respectivement les états de liquide de saturation et de vapeur saturante. Les états métastables de liquide sursaturé $L_s L'_s$ et de vapeur sursaturée $V_s V'_s$ sont effectivement observés.



Que peut-on dire des valeurs μ_ℓ et μ_v des fonctions de Gibbs molaires (ou potentiels chimiques) des deux phases dans les états de saturation ?

Dans les états de saturation, l'équilibre des deux phases sous la pression $p = p_s$ se traduit par l'égalité de leurs potentiels chimiques respectifs : $\mu_\ell(T, p_s) = \mu_v(T, p_s)$.

Montrer que le volume molaire est une dérivée partielle du potentiel chimique et en déduire que le palier de changement d'état doit être placé de telle sorte que les aires algébriques $A_\ell (< 0)$ et $A_v (> 0)$ soient opposées (construction de Maxwell : $A_\ell + A_v = 0$).

Partant de l'identité thermodynamique molaire, nous obtenons successivement :

$$dU_m = -p dV_m + T dS_m$$

$$dH_m = d(U_m + pV_m) = -p dV_m + T dS_m + d(pV_m) = +V_m dp + T dS_m$$

$$dG_m = d(H_m - TS_m) = V_m dp + T dS_m - d(TS_m) = V_m dp - S_m dT$$

$$\text{Nous en déduisons : } d\mu = dG_m = V_m dp - S_m dT = \left(\frac{\partial \mu}{\partial p}\right)_T dp + \left(\frac{\partial \mu}{\partial T}\right)_p dT$$

$$\text{Soit, en particulier, par identification : } V_m = \left(\frac{\partial \mu}{\partial p}\right)_T \quad \text{c.q.f.d.}$$

L'intégrale de la fonction $p(V_m)$ du point L_s au point V_s correspond à la somme de l'aire algébrique recherchée A et de l'aire $p_s(V_{mvs} - V_{m\ell s})$ sous le palier de changement d'état.

$$\int_{L_s V_s} p dV_m = A + p_s (V_{mvs} - V_{m\ell s})$$

Cette intégrale peut s'exprimer par une intégration par parties faisant apparaître $\mu_v - \mu_\ell$:

$$\int_{L_s V_s} p dV_m = [pV_m]_{L_s}^{V_s} - \int_{L_s V_s} V_m dp = p_s (V_{mvs} - V_{m\ell s}) - \int_{L_s V_s} \left(\frac{\partial \mu}{\partial p}\right)_T dp = p_s (V_{mvs} - V_{m\ell s}) - (\mu_v - \mu_\ell)$$

$$\text{Nous en déduisons : } A = A_\ell + A_v = \mu_\ell - \mu_v = 0 \quad \text{c.q.f.d.}$$

- 4- À la température $t = 25^\circ\text{C}$, la pression du changement d'état *liquide* \rightarrow *vapeur* a pour valeur $p_s = 64,3$ bar. Cette valeur est-elle compatible avec le modèle de Van der Waals déterminé à la question 1 ?

Nous pouvons lire sur le réseau d'isothermes précédent que pour des pressions inférieures à 67 bar, il n'existe qu'un seul volume possible. La pression de vapeur saturante à 25°C est bien trop petite pour être compatible avec le modèle de Van der Waals paracritique.

Posons la question poliment à Maple. La résolution d'une équation du troisième degré ne pose aucun problème et la machine nous renvoie deux valeurs complexes conjuguées et une valeur réelle :

> **t:=25:**

ps:=64.3e5:

solve(p=ps,v);

0.0000757746-0.0000119899 I, 0.0000757746+0.0000119899 I, 0.000162192

Conclusion ?

Le modèle de Van der Waals paracritique ne convient donc plus si l'on s'éloigne trop de l'état critique.

Deuxième partie :**Modèle de Van der Waals optimisé pour la température de 25°C au voisinage du palier de changement d'état**

5- En réalité, les volumes molaires du liquide de saturation et de la vapeur saturante à 25°C sont respectivement $V_{mLs} = 62 \text{ mL} \cdot \text{mol}^{-1}$ et $V_{mVs} = 184 \text{ mL} \cdot \text{mol}^{-1}$. Bien que nous ne soyons pas très loin du point critique, ces valeurs sont incompatibles avec le modèle de Van der Waals étudié à la question 1.

Déterminer la valeur des paramètres a , b et r du gaz de Van der Waals ayant le même palier de changement d'état que le dioxyde de carbone à 25°C et respectant la construction de Maxwell.

Indication :

i- Définir l'intégrale $A = \int_{V_{mLs}}^{V_{mVs}} (p - p_s) dV_m$ comme expression formelle de r , b et a .

```
> p:=r*T/(v-b)-a/v**2;
T:=25+273.15;
vls:=62e-6; vvs:=184e-6;
ps:=64.3e5;
A:=int(p-ps,v=vls..vvs):
```

$$p := \frac{r T}{v - b} - \frac{a}{v^2}$$

ii- Déterminer le paramètre r comme expression formelle de a et b (procédure Maple **solve**) telle que l'isotherme passe par le point $L_s(V_{mLs}, p_s)$. L'aire A est alors une expression des deux variables formelles a et b .

```
> v:=vvs;r:=solve(p=ps,r);
```

$$r := 18.22835019 a - 99067.12058 a b + 3.968203924 - 21566.32567 b$$

iii- Déterminer le paramètre b comme expression formelle de a (procédure Maple **solve**) telle que l'isotherme passe par le point $L_v(V_{mVs}, p_s)$. L'aire A est alors une expression de la seule variable formelle a .

```
> v:=vls;b:=solve(p=ps,b);
```

$$b := \frac{0.0002000000000 (0.1069424965 \cdot 10^{15} a - 0.7844599999 \cdot 10^{13})}{0.4612176392 \cdot 10^{15} a + 2979.}$$

iv- Déterminer numériquement a (procédure Maple **fsolve**) de telle sorte que $A(a) = 0$.

```
> unassign('v');
a:=fsolve(A=0,a);
b:=b;r:=r;
```

$$a := 0.2329120367$$

$$b := 0.00003176893678$$

$$r := 6.795632796$$

Remarque : Les nouvelles valeurs sont du même ordre de grandeur que les valeurs « paracritiques »


$a = 0,233 \text{ Pa} \cdot \text{m}^6 \cdot \text{mol}^{-2}$ au lieu de $a = 0,196 \text{ Pa} \cdot \text{m}^6 \cdot \text{mol}^{-2}$,

$b = 31,8 \times 10^{-6} \text{ m}^3 \cdot \text{mol}^{-1}$ au lieu de $b = 31,3 \times 10^{-6} \text{ m}^3 \cdot \text{mol}^{-1}$,

$r = 6,80 \text{ J} \cdot \text{K}^{-1} \cdot \text{mol}^{-1}$ au lieu de $r = 6,09 \text{ J} \cdot \text{K}^{-1} \cdot \text{mol}^{-1}$

6- Représenter sur un même graphe la branche liquide et la branche vapeur de l'isotherme de Van der Waals à 25°C ainsi que le palier de changement d'état correspondant. Superposer, par une courbe en style pointillé —option **linestyle=3**—, les états instables prévus pour cette température par le modèle de Van der Waals.

```
> branche_liquide:=plot(p,v=0..vls,y=0.8*ps..1.2*ps):
palier_de_vaporisation:=plot(ps,v=vls..vvs):
branche_vapeur:=plot(p,v=vvs..1.2*vvs):
etats_instables:=plot(p,v=vls..vvs, linestyle=3, thickness=1):
plots[display](branche_liquide,
                palier_de_vaporisation,
                branche_vapeur,
                etats_instables,
                thickness=3, labels=['Vm', 'p']);
```



Remarque : sur le graphe, tracé par Maple, les indications en bleu ont été ajoutées pour plus de clarté.

7- Montrer que le potentiel chimique $\mu(p)$ d'un fluide réel est une fonction continue présentant une discontinuité de dérivée pour $p = p_s$.

Le potentiel chimique est une fonction continue de p pour chaque phase. La condition d'équilibre thermodynamique du système diphasé, $\mu_\ell(p_s) = \mu_v(p_s)$, assure la continuité pour $p = p_s$.

À température constante, la dérivée du potentiel chimique par rapport à la pression est égale au volume molaire.


La discontinuité du volume molaire à la transition de phase, $V_{m\ell s} < V_{mv s}$, implique la discontinuité de la dérivée de la fonction $\mu(p)$:

$$\lim_{p \rightarrow p_s^+} \left(\frac{\partial \mu}{\partial p} \right)_T = V_{m\ell s} < \lim_{p \rightarrow p_s^-} \left(\frac{\partial \mu}{\partial p} \right)_T = V_{mv s}$$

8- À l'aide d'un logiciel de calcul formel, représenter le graphe $p(\mu)$ pour le dioxyde de carbone à la température $t = 25^\circ\text{C}$ dans le cadre du modèle de Van der Waals étudié à la question 4.

Indication : $p(V_m)$ et $\mu(V_m)$ sont des fonctions. Le graphe $p(\mu)$ quant à lui ne représente pas une fonction (en effet, plusieurs valeurs de μ peuvent correspondre à une même valeur de p). Ce graphe doit être construit comme une courbe paramétrique $[\mu(V_m), p(V_m)]$. Voir `?plot[parametric]` avec l'aide de Maple.

```
> mu:=p*v-int(p,v):
v:=vls:mus:=mu:v:='v':
mu:=mu-mus:
mu_liquide:=plot([mu(v),p(v),v=0.8*vls..vls],
-100..100,0.8*ps..1.2*ps):
mu_instable:=plot([mu(v),p(v),v=vls..vvs],
-100..100,0.8*ps..1.2*ps,linestyle=3,thickness=1):
mu_vapeur:=plot([mu(v),p(v),v=vvs..1.2*vvs],
-100..100,0.8*ps..1.2*ps):
plots[display](mu_liquide,
mu_instable,
mu_vapeur,
thickness=3,labels=['mu','p']);
```



D'après l'allure de ce graphe, peut-on prévoir l'instabilité des états sursaturés ?

Nous constatons, pour une pression donnée, qu'à chaque état sursaturé correspond un état thermodynamiquement plus stable dont le potentiel chimique est moindre. Les états sursaturés tels que L' sont donc métastables.

Nous pouvons les observer parce que la transition *état sursaturé* \rightarrow *état stable* nécessite une certaine activation.

Par contre, entre les deux points de rebroussement, les états prédits par l'équation de Van der Waals, de potentiels chimiques encore plus élevés que les états sursaturés, ne correspondent à aucune réalité.